

メモ

例題の定義

0.1 例題の定義

0.1.1 LMNtal プログラム

プログラムには, 初期状態として, 1 つのハイパーリンクを引数として持つアトム `start_1` 1 つと, 整数, ハイパーリンク, ハイパーリンク, 整数の組を引数として持つアトム `distance_4` がいくつか与えられる.

```

1 start(!X) :- between(0,!X,0).
2
3 max@@ R = max(A,B) :- A > B | R = A.
4 max@@ R = max(A,B) :- A =< B | R = B.
5 min@@ R = min(A,B) :- A > B | R = B.
6 min@@ R = min(A,B) :- A =< B | R = A.
7
8 inconsistent@@
9 between(A,!X,B) :- A > B | fail(A,!X,B).
10
11 intersect@@
12 between(A,!Y,B), between(C,!Y,D) :- between(max(A,C),!Y,min(B,D)).
13
14 propagate_forward@@
15 between(A,!Y,B), distance(C,!Y,!Z,D) \
16     :- AC=A+C, BD=B+D, uniq(A,!Y,B,C,!Z,D)
17     | between(AC,!Z,BD).
18
19 propagate_backward@@
20 between(A,!Y,B), distance(C,!Z,!Y,D) \
21     :- AC=A-D, BD=B-C, uniq(A,!Y,B,C,!Z,D)
22     | between(AC,!Z,BD).

```

0.1.2 各アトムとプログラムの数学としての意味

プログラム内には, `between_3`, `distance_4` が, 主要なアトムとして出現し, 引数に含まれるハイパーリンクは, 整数上の変数を意味する. `between_3` の引数は, 整数 a , ハイパーリンク H , 整数 b の組であり, $a \leq H \leq b$ を意味する. `distance_4` の引数は, 整数 a , ハイパーリンク H_1 , ハイパーリンク H_2 , 整数 b の組であり, $a \leq H_2 - H_1 \leq b$ を意味する.

$\min \leq H_1 \leq \max$ かつ $a \leq H_2 - H_1 \leq b$ であるとき, $a + \min \leq H_2 \leq b + \max$ がわかる. また, $\min \leq H_2 \leq \max$ かつ $a \leq H_2 - H_1 \leq b$ であるとき, $\max - b \leq H_1 \leq \min - a$ がわかる.

ある変数の集合 V を考える. 今, $0 \leq H_0 \leq 0$ where $H_0 \in V$ を表す `between_3` アト

ムが 1 つと , $a \leq H_j - H_i \leq b$ where $H_i \in V, H_j \in V, a \in \mathbb{Z}, b \in \mathbb{Z}$ を表す distance4
アトムが複数与えられるとする .

このとき , 計算可能な全ての変数 $H \in V$ について , $\min \leq H \leq \max$ を計算する .

0.1.3 このプログラムの実用例題的な意味 (問題文)

あなたは部署の依存関係のあるタスクの時間管理を任されました . あなたには , 2 つの
タスク間の , Task₂ が必ず Task₁ の a 分後から b 分後までの間に実行されるという関係
が複数知らされます . また , どのタスクが始めに行われるべきかということが知らされま
す . 始めに行われるべきタスクの時間をタスクの開始後 0 分後 から 0 分後であるとする
と , 各タスクは開始後何分後から何分後に実行されるでしょうか .